Einbinden von FLAM® in Visual Basic (.NET)
(Diese Beschreibung gilt sinngemäß auch für andere .NET-Programmiersprachen)

FLAM® kann anstelle der Eingabedatei den Standardeingangskanal benutzen (StdIn), sowie für die Ausgabedatei den Standardausgangskanal (StdOut). Die Bildschirmmeldungen schickt FLAM® über den Standardfehlerkanal (StdErr).

Programme, die sich in diese Kanäle einschalten, können so FLAM® direkt steuern.

In Visual Basic geht das zum Beispiel so (es geht auch anders):

1. Am Besten Systemvariable setzen mit der Position von FLAM®
(wie FLAMFILE = C:\Program Files\FLAM4\Flam4.EXE).

2. Klasse FLAMTasks schreiben (oder kopieren):

Imports System.Diagnostics
Imports System.IO

Imports System.Threading

Friend Class FLAMTasks

    Public FLAMCommand As String

    Public FLAMParameters As String

    Public FLAMIn As String

    Public FLAMOut As String

    Dim FLAMProcess As Process = New Process

    Public Sub FLAMRun() 'startet FLAM®

        With FLAMProcess

            With .StartInfo

                .FileName = FLAMCommand

                .Arguments = FLAMParameters

                .UseShellExecute = False

                .CreateNoWindow = True

                .RedirectStandardError = True

                .RedirectStandardInput = True

                .RedirectStandardOutput = True

            End With

            .Start()

            Dim readThread As New Thread(AddressOf FLAMRead)

            readThread.Start()

            Dim writeFileThread As New Thread(AddressOf FLAMWriteFile)

            writeFileThread.Start()

            Dim readFileThread As New Thread(AddressOf FLAMReadFile)

            readFileThread.Start()

        End With

    End Sub

    Private Sub FLAMRead() 'liest FLAM®-Ausgaben

        Dim sErr As StreamReader = FLAMProcess.StandardError

        Dim line As String = ""

        With sErr

            Do Until FLAMProcess.StandardError.EndOfStream

                FLAMProcess.Refresh()

                line = .ReadLine
'hier könnte man was mit der Ausgabe von FLAM® machen.
                Thread.Sleep(0)

            Loop

            .Close()

        End With

    End Sub

    Private Sub FLAMWriteFile() 'füttert FLAM®

        Dim bufferSize As Integer = 4096 'zum Beispiel

        Dim buffer(bufferSize) As Byte

        Dim writer As BinaryWriter = New BinaryWriter(FLAMProcess.StandardInput.BaseStream)

        Dim reader As BinaryReader = Nothing

        Try

            reader = New BinaryReader(File.Open(FLAMIn, FileMode.Open))

            Do

                buffer = reader.ReadBytes(bufferSize)

                Dim l As Integer = buffer.Length

                If l <= 0 Then

                    Exit Do

                End If

                writer.Write(buffer, 0, l)
'hier könnte man einen Fortschrittsbalken weiterführen.
                Thread.Sleep(0)

            Loop

        Catch ex As Exception

            Debug.Write(ex.Message)

        Finally

            Try

                writer.Flush()

                writer.Close()

                reader.Close()

            Catch ex As Exception

                Debug.Write(ex.Message)

            End Try

        End Try

    End Sub

    Private Sub FLAMReadFile() 'melkt FLAM®

        Dim reader As BinaryReader = New BinaryReader(FLAMProcess.StandardOutput.BaseStream)

        Dim bufferSize As Integer = 4096 'zum Beispiel

        Dim buffer(bufferSize) As Byte

        Dim writer As BinaryWriter = Nothing

        Try

            writer = New BinaryWriter(File.Open(FLAMOut, FileMode.Create))

            Do

                buffer = reader.ReadBytes(bufferSize)

                Dim l As Integer = buffer.Length

                If l = 0 Then

                    Exit Do

                End If

                writer.Write(buffer, 0, l)

                Thread.Sleep(0)

            Loop

        Catch ex As Exception

            Debug.Write(ex.Message)

        Finally

            Try

                reader.Close()

                writer.Flush()

                writer.Close()

            Catch ex As Exception

                Debug.Write(ex.Message)

            End Try

        End Try

    End Sub

End Class
3. Startkommando und gewünschte Parameter zusammenbasteln und die FLAMTasks aufrufen:

Imports System.Environment
Imports System.Threading

Public Class FLAMExample

    Private Sub Main()

        Dim FLAMIn As String = Nothing

        Dim FLAMOut As String = Nothing

        FLAMIn = "EingabeDatei.xxx"
        FLAMOut = "DaSollsHin.yyy”
        Dim FLAMTasks As New FLAMTasks

        Dim FLAMThread As New Thread(AddressOf FLAMTasks.FLAMRun)

        With FLAMTasks

            .FLAMCommand = GetEnvironmentVariable("FLAMFILE")

            .FLAMParameters = "-compress -adc -cryptmode=aes -password=GEHEIM -secureinfo=yes" 'zum Beispiel

            .FLAMIn = FLAMIn

            .FLAMOut = FLAMOut

        End With

        FLAMThread.Start()

    End Sub

End Class

4. Fertig.
